

Rules for integrands of the form $(a + b \operatorname{ArcCosh}[c x])^n$

1: $\int (a + b \operatorname{ArcCosh}[c x])^n dx$ when $n > 0$

Derivation: Integration by parts

Basis: $\partial_x (a + b \operatorname{ArcCosh}[c x])^n = \frac{b c n (a + b \operatorname{ArcCosh}[c x])^{n-1}}{\sqrt{1+c x} \sqrt{-1+c x}}$

Rule: If $n > 0$, then

$$\int (a + b \operatorname{ArcCosh}[c x])^n dx \rightarrow x (a + b \operatorname{ArcCosh}[c x])^n - b c n \int \frac{x (a + b \operatorname{ArcCosh}[c x])^{n-1}}{\sqrt{1+c x} \sqrt{-1+c x}} dx$$

Program code:

```
Int[(a_ + b_.*ArcCosh[c_.*x_])^n_., x_Symbol] :=  
  x*(a+b*ArcCosh[c*x])^n -  
  b*c*n*Int[x*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]), x] /;  
FreeQ[{a,b,c}, x] && GtQ[n, 0]
```

$$2: \int (a + b \operatorname{ArcCosh}[c x])^n dx \text{ when } n < -1$$

Derivation: Integration by parts

$$\text{Basis: } \frac{(a+b \operatorname{ArcCosh}[c x])^n}{\sqrt{1+c x} \sqrt{-1+c x}} == \partial_x \frac{(a+b \operatorname{ArcCosh}[c x])^{n+1}}{b c (n+1)}$$

Rule: If $n < -1$, then

$$\int (a + b \operatorname{ArcCosh}[c x])^n dx \rightarrow \frac{\sqrt{1+c x} \sqrt{-1+c x} (a + b \operatorname{ArcCosh}[c x])^{n+1}}{b c (n+1)} - \frac{c}{b (n+1)} \int \frac{x (a + b \operatorname{ArcCosh}[c x])^{n+1}}{\sqrt{1+c x} \sqrt{-1+c x}} dx$$

Program code:

```
Int[(a_+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1+c*x]*Sqrt[-1+c*x]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c/(b*(n+1))*Int[x*(a+b*ArcCosh[c*x])^(n+1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c},x] && LtQ[n,-1]
```

$$3: \int (a + b \operatorname{ArcCosh}[c x])^n dx$$

Derivation: Integration by substitution

Basis:

$$(a + b \operatorname{ArcCosh}[c x])^n = \frac{1}{bc} \operatorname{Subst}\left[x^n \operatorname{Sinh}\left[-\frac{a}{b} + \frac{x}{b}\right], x, a + b \operatorname{ArcCosh}[c x]\right] \partial_x (a + b \operatorname{ArcCosh}[c x])$$

Rule:

$$\int (a + b \operatorname{ArcCosh}[c x])^n dx \rightarrow \frac{1}{bc} \operatorname{Subst}\left[\int x^n \operatorname{Sinh}\left[-\frac{a}{b} + \frac{x}{b}\right] dx, x, a + b \operatorname{ArcCosh}[c x]\right]$$

Program code:

```
Int[(a_ + b_.*ArcCosh[c_.*x_])^n_, x_Symbol] :=
  1/(b*c)*Subst[Int[x^n*Sinh[-a/b+x/b], x], x, a+b*ArcCosh[c*x] ] /;
FreeQ[{a, b, c, n}, x]
```